

Lesson 42. A Complex Interferometer

This lesson goes through the steps of setting up an interferometer.

Interferometers have two channels, and the beams are combined at a beamsplitter. One often wants to see the difference in the shape of the two wavefronts, as when testing aspheric mirrors.

In this example, fringes between the two channels give spectral information as the position of one of the mirrors is moved back and forth. In this configuration, the instrument is called a *Fourier-Transform* Interferometer. Here one is not concerned with the shape of the wavefront, but with its absolute phase.

We will set up one channel at first, entering those data that are easy to figure out, and then let the program calculate the rest for us. Here is the input for the first step:

```
RLE
ID INTERFEROMETER EXAMPLE
WAVL 4.6 4.25 3.9
OBB 0 1 30 0 0 0 30
1 TH 100          ! DUMMY SURFACE FOR REFERENCE
2 AT -45 0 100    ! SCAN MIRROR
2 REFL
2 TH 0
3 AT -45 0 100    ! FOLD AXIS
3 TH -200         ! TO BEAMSPLITTER
4 TH -3 GTB U     ! THROUGH 3 MM THICK GERMANIUM
GE
5 REFL           ! REFLECT AT BEAMSPLITTER
5 PTH -4 PIN 4   ! COMING BACK AGAIN
4 AT 30 0 100    ! TILT OF BEAMSPLITTER
7 AT 30 0 100    ! TILT AXIS
7 TH 70          ! TO REFERENCE MIRROR
8 REFL           ! HERE
8 PTH -7         ! BACK TO BEAMSPLITTER

9 AT -30 0 100   ! ENTER IT AGAIN
9 PTH 4 PIN 4    ! SAME SIZE AS BEFORE
10 TH -.1        ! SMALL AIRSPACE
11 PTH 4 PIN 4   ! COMPENSATOR DUPLICATES GEOMETRY
12
13 AT 30 0 100   ! DUMMY TO FOLLOW BEAM
13 TH -200       ! DISTANCE TO FOLD MIRROR 1
14 AT -30 0 100  ! RIGHT HERE
14 GID
FOLD 1          ! IDENTIFY IT
14 REFL         ! REFLECT THERE
15 AT -30 0 100 ! DUMMY TO FOLLOW BEAM
15 TH 250       ! DISTANCE TO PRIMARY MIRROR
16 RD -180 CC -1 TH -90 ! PARABOLOID HERE
16 REFL         ! REFLECT THERE
16 GID
PRIMARY M      ! IDENTIFY IT
17 AT 45 0 100  ! SMALL FOLD MIRROR
17 REFL         ! REFLECT HERE
17 GID
SECONDARY M    ! IDENTIFY IT
18 AT 45 0 100  ! DUMMY TO FOLLOW BEAM
18 TH 90        ! DISTANCE TO TERTIARY
19 RD -180 TH -350 ! DISTANCE TO FINAL IMAGE
19 REFL         ! REFLECT AT TERTIARY
```


```

APS 19
19 GID
TERTIARY M
20
END

```

! THE STOP IS HERE

We type the above into the EE editor and run it, giving this picture in PAD:



To get this display, we click on the PAD Top button, select Custom rayset, HBAR 0.0 and 11 rays. We also select the Solo top display option and turn on switch 38, which shows numbers for all surfaces, including dummies.


So far, we have the basic elements in place, but we do not yet know the details of the tertiary mirror. We want a sharp image on surface 20, and we will insert additional fold mirrors to separate three wavelength regions onto different detectors later when we get to that step. First, we need to know the radius and conic constant on 19. We type the following into a new editor:

```

PANT
VY 19 ASPH
END
AANT
GSR 0 1 4 P
END

```

After running this file, our system looks just the way we want it to:



The command ASY now shows us the shape of surface 19:

SYNO PSYS AI>ASY

SPECIAL SURFACE DATA

SURFACE NO. 16 -- CONIC SURFACE
 CONIC CONSTANT (CC) -1.000000
 SEMI-MAJOR AXIS (b) INFINITE SEMI-MINOR AXIS (a) INFINITE

SURFACE NO. 19 -- CONIC SURFACE
 CONIC CONSTANT (CC) -0.349174
 SEMI-MAJOR AXIS (b) -219.999999 SEMI-MINOR AXIS (a) 177.482393


TILT AND DECENTER DATA
LEFT-HANDED COORDINATES

SURF TYPE	X	Y	Z	ALPHA	BETA	GAMMA
-----------	---	---	---	-------	------	-------

2 REL	0.00000	0.00000	0.00000	-45.0000	0.0000	0.0000
3 REL	0.00000	0.00000	0.00000	-45.0000	0.0000	0.0000
4 REL	0.00000	0.00000	0.00000	30.0000	0.0000	0.0000
7 REL	0.00000	0.00000	0.00000	30.0000	0.0000	0.0000
9 REL	0.00000	0.00000	0.00000	-30.0000	0.0000	0.0000
13 REL	0.00000	0.00000	0.00000	30.0000	0.0000	0.0000
14 REL	0.00000	0.00000	0.00000	-30.0000	0.0000	0.0000
15 REL	0.00000	0.00000	0.00000	-30.0000	0.0000	0.0000
17 REL	0.00000	0.00000	0.00000	45.0000	0.0000	0.0000
18 REL	0.00000	0.00000	0.00000	45.0000	0.0000	0.0000

KEY TO SURFACE TYPES

GLB	GLOBAL COORDINATES	LOC	LOCAL COORDINATES
REL	RELATIVE COORDINATES	REM	REMOTE TILTS IN RELATIVE COORD.
SYNOPTSYS AI>			


Okay, one channel looks good; now let's set up the second. We can start with the above setup and just modify it as required. First, we bump this setup into ACON 2, with the ACON copy button , and then we modify the geometry at the beamsplitter. We make a CHG file: (L42M2)

```

CHG
13 SIN          ! NEED TWO ADDITIONAL SURFACES
13 SIN
4 NAS          ! REMOVE TILTS THERE NOW
7 NAS
9 NAS
13 NAS
4 TH -3 GTB U
GE
4 AT 30 0 100  ! TILT OF BEAMSPLITTER
5 TH -.1 TRANS
6 PTH 4 PIN 4
7 TH 0
8 AT -30 0 100
8 TRANS       ! NOT REFLECTIVE ANYMORE
8 TH -70
9 REFL
9 PTH -8 AIR
10 AT 30 0 100
10 PTH -4 PIN 4
11 PTH -5
12 REFL
12 PTH 5
13 PTH 4 PIN 4
14 TH 0
15 AT 30 0 100
15 TH -200
END

```

In this file, we first have to delete most of the declarations assigned to the beamsplitter in channel 1, since reflections and tilts now occur on different surfaces, and then we replace them with the data for the other channel. The new system looks like this:



(Here we have turned off switch 38 to make a cleaner picture.)

Okay, we have both channels defined, and they are both current, in ACONS 1 and 2. Now we can make a perspective drawing showing both at the same time. We create a MACro:

```

ACON 1
HPLOT 1
PER 0 0 0 1 123
PUP 2 1 11
PLOT
RED
TRA P 0 0 11
END


```

```

ACON 2
APLOT 1
PER 0 0 0 1 123
PUP 2 1 11
PLOT
BLUE
TRA P 0 0 11
END

```

This gives us the picture below:



Not bad. Let's improve it some more. We open the Edge Wizard (MEW) and select Create All. Do this for both ACONs. Now run the MACro above, after turning on switch 20, adding TRA requests for HBAR = 1 and -1, and changing the PER requests to RSOLID. The picture is below. We are off to a very good start. (L42L1)

This is a very brief lesson on how to set up systems like this. SYNOPSYS can show the system and the image quality very nicely – **but it will not model the interference between the two channels**. The program can model the difference in *shape* between a wavefront and a reference beam (see **HELP IFR**), but the information in a Fourier-transform interferometer arises from the absolute *phase* of the wavefronts, which is not calculated by SYNOPSYS. Of course, this is no matter since the theory is already very mature, and all we care about is image quality in these systems.

The next step will be to add additional fold mirrors and detector optics in the space before the final image. But now that you know how these things are done, we leave that as an exercise for the student.

If you are especially perceptive, you will have noticed that there is a small decenter in the beam as it goes through the beamsplitter – which we have ignored. But that's not hard to model either; just adjust a decenter on the primary mirror to compensate, if you really want to get that precise.

